Supplemental Material

Appendix: This appendix documents additional information for readers who are interested in technical details.

Supplemental Table 1. Definitions, mathematical forms and relationships among four functions regarding time to event in survival analyses (Pr: probability function; T: survival time, a random variable; t: a specific time point; Δt: a period of time; d: derivative; log: function of logarithm; ∫: integration)

Functions of Time	Definition	Mathematical Form
Survival, $S(t)$	The probability of an individual being free from an event beyond certain time point *t*	$S(t) = \Pr(T > t)$
Cumulative incidence, $F(t)$	The probability of an individual experiencing an event before time *t*	$F(t) = \Pr(T \leq t) = 1 - S(t)$
Hazard, $h(t)$	The instantaneous (such that $\Delta t \rightarrow 0$) rate where the event occurs for an individual surviving at time *t*	$h(t) = \lim_{\Delta t \rightarrow 0} \frac{\Pr(t < T \leq t + \Delta t \mid T > t)}{\Delta t} = \frac{f(t)}{S(t)} = \frac{dF(t)/dt}{S(t)} - \frac{d\log\{S(t)\}}{dt}$
Cumulative hazard, $H(t)$	The accumulation of hazard rate over time	$H(t) = \int_0^t h(u)du = \int_0^t -d\log\{S(u)\} = -\log\{S(t)\}$

Supplemental material is neither peer-reviewed nor thoroughly edited by CJASN. The authors alone are responsible for the accuracy and presentation of the material.

Supplemental Table 2: Analyses using one or both of cause-specific and sub-distribution hazards models for one or both of the event of interest and the competing event and the corresponding restrictions on hazard ratios. Note that Analyses #1 is of interest in most situations, including examples demonstrated in this manuscript. ($h_1(t|\boldsymbol{X})$: cause-specific hazards rate for event of interest (subscript of 2 for competing event) conditional on covariates $\boldsymbol{X}$; $\lambda_1(t|\boldsymbol{X})$: sub-distribution hazards rate for event of interest (subscript of 2 for competing event) conditional on covariates $\boldsymbol{X}$)

Analyses	Cause-specific hazards models		Sub-distribution hazards models		Restrictions
	Event of interest (1)	Competing event (2)	Event of interest (1)	Competing event (2)	
1	X		X		$\mathrm{HR}_{1,\mathrm{CS}} = \mathrm{SHR}_1 \leq 1 + \min\left[\frac{\{1-F_1(t\|\boldsymbol{X}^*)\}F_2'(t\|\boldsymbol{X}^*)}{F_1'(t\|\boldsymbol{X}^*)F_2(t\|\boldsymbol{X}^*)}\right]$
2		X		X	$\mathrm{HR}_{2,\mathrm{CS}} = \mathrm{SHR}_2 \leq 1 + \min\left[\frac{\{1-F_2(t\|\boldsymbol{X}^*)\}F_2'(t\|\boldsymbol{X}^*)}{F_2'(t\|\boldsymbol{X}^*)F_2(t\|\boldsymbol{X}^*)}\right]$
3			X	X	$\mathrm{SHR}_1 = \log\{\mathrm{P}(D=2\|\boldsymbol{X})\}/\log\{\mathrm{P}(D=2\|\boldsymbol{X}^*)\}$ and $\mathrm{SHR}_2 = \log\{\mathrm{P}(D=1\|\boldsymbol{X})\}/\log\{\mathrm{P}(D=1\|\boldsymbol{X}^*)\}$. If $\mathrm{P}(D=1\|\boldsymbol{X}) > \mathrm{P}(D=1\|\boldsymbol{X}^*)$, $\mathrm{SHR}_1 > 1$ and $\mathrm{SHR}_2 < 1$; if $\mathrm{P}(D=1\|\boldsymbol{X}) < \mathrm{P}(D=1\|\boldsymbol{X}^*)$, $\mathrm{SHR}_1 < 1$ and $\mathrm{SHR}_2 < 1$; if $\mathrm{P}(D=1\|\boldsymbol{X}) = \mathrm{P}(D=1\|\boldsymbol{X}^*)$, $\mathrm{SHR}_1 = \mathrm{SHR}_2 = 1$
4	X			X	$\mathrm{HR}_{1,\mathrm{CS}}$and SHR_2 are inversely related in a weaker fashion
5		X	X		SHR_1 and $\mathrm{HR}_{2,\mathrm{CS}}$ are inversely related in a weaker fashion
6	X	X	X		Under three proportionality assumptions, $\mathrm{HR}_{1,\mathrm{CS}} = \mathrm{SHR}_1$ and then $h_1(t\|\boldsymbol{X}) = \lambda_1(t\|\boldsymbol{X})$, which is not possible because $h_1(t\|\boldsymbol{X})$ is greater than $\lambda_1(t\|\boldsymbol{X})$ by definition
7	X		X	X	Under three proportionality assumptions, $\mathrm{HR}_{2,\mathrm{CS}} = \mathrm{SHR}_2$ and then $h_2(t\|\boldsymbol{X}) = \lambda_2(t\|\boldsymbol{X})$, which is not possible because $h_2(t\|\boldsymbol{X})$ is greater than $\lambda_2(t\|\boldsymbol{X})$ by definition
8	X	X		X	Under three proportionality assumptions, $\mathrm{HR}_{1,\mathrm{CS}} = \mathrm{SHR}_1$ and then $\mathrm{SHR}_2 = 1$, which is not possible because SHR_1 and SHR_2 have to be on opposite side of 1
9		X	X	X	Under three proportionality assumptions, $\mathrm{HR}_{2,\mathrm{CS}} = \mathrm{SHR}_2$ and then $\mathrm{SHR}_1 = 1$, which is not possible because SHR_1 and SHR_2 have to be on opposite side of 1

```
* Create a dataset containing a set of predictors at baseline level for which the quantities of interest (e.g., baseline
cumulative hazard or baseline cumulative incidence functions) are estimated;
data risk;
  predictor = 0;
run;

***** Cause-specific (CS) hazards models *****;
proc phreg data = dataset;
  class predictor; * for categorical variables;
  * 1: event of interest, 0: censoring events, 2: competing event;
  model time*event(0,2) = predictor / risklimits;
  baseline covariates=risk out=outdata1 cumhaz=_all_ / method=breslow; * save baseline cumulative hazard functions (CIFs) at
each time point in outdata1;
run;
* To calculate CIFs, we need to fit CS hazards models for the competing event;
proc phreg data = dataset;
  class predictor; * for categorical variables;
  * 1: event of interest, 0: censoring events, 2: competing event;
  model time*event(0,1) = predictor / risklimits;
  baseline covariates=risk out=outdata2 cumhaz=_all_ / method=breslow;
run;
* Calculate CS-based CIFs for baseline group;
data cs_cif_baseline;
  merge outdata1(keep = time cumhaz rename = (cumhaz = _cumhaz1))
        outdata2(keep = time cumhaz rename = (cumhaz = _cumhaz2));
  by time;
  retain cumhaz1 cumhaz2;
  if not missing(_cumhaz1) then cumhaz1 = _cumhaz1;
  if not missing(_cumhaz2) then cumhaz2 = _cumhaz2;
  survival = exp(-(cumhaz1 + cumhaz2)); * Calculate overall baseline survival function;
  hazard1 = cumhaz1 - lag(cumhaz1); * Calculate baseline CS hazard function;
  hazard2 = cumhaz2 - lag(cumhaz2); * Calculate baseline CS hazard function;
  haz1_surv = hazard1*survival; * product of CS hazard and overall survival functions;
  haz2_surv = hazard2*survival; * product of CS hazard and overall survival functions;
  if _n_ = 1 then do;
    hazard1 = 0;
    hazard2 = 0;
    cif1 = 0;
    cif2 = 0;
  end;
  cif1 + haz1_surv; * Integrating the above product over time;
  cif2 + haz2_surv; * Integrating the above product over time;
  drop _cumhaz1 _cumhaz2 haz1_surv haz2_surv;
run;
* Calculate CS-based CIFs for the other group;
data cs_cif;
  set cs_cif_baseline(keep = time hazard1 hazard2
                      rename = (hazard1 = hazard1_baseline hazard2 = hazard2_baseline));
  hazard1 = hazard1_baseline*&HR1; * Extra effort is required to save CS hazard ratio for event of interest into a SAS macro
variable HR1;
  hazard2 = hazard2_baseline*&HR2; * HR2 is a SAS macro variable for CS hazard ratio for the competing event;
  if _n_ = 1 then do;
    cumhaz1 = 0;
    cumhaz2 = 0;
  end;
  cumhaz1 + hazard1; * Calculate cumulative hazard function;
  cumhaz2 + hazard2; * Calculate cumulative hazard function;
  survival = exp(-(cumhaz1 + cumhaz2)); * Calculate overall survival function;
  haz1_surv = hazard1*survival; * product of CS hazard and overall survival functions;
  haz2_surv = hazard2*survival; * product of CS hazard and overall survival functions;
  if _n_ = 1 then do;
    cif1 = 0;
    cif2 = 0;
  end;
  cif1 + haz1_surv; * Integrating the above product over time;
  cif2 + haz2_surv; * Integrating the above product over time;
  drop haz1_surv haz2_surv;
run;

***** Sub-distribution (SD) hazards models *****;
proc phreg data = dataset;
  class predictor; * for categorical variables;
  * 1: event of interest, 0: censoring events, 2: competing event;
  * eventcode=1 tells SAS 1 is the event of interest and 2 is the competing event;
  model time*event(0) = predictor / eventcode=1 risklimits;
  baseline covariates=risk out=outdata cif = _all_ / seed=19104; * save baseline CIFs at each time point in outdata;
run;
* Calculate SD-based CIFs for baseline group;
data sd_cif_baseline;
  set outdata(keep = time cif rename = (cif = cif1));
  cumhaz1 = -log(1 - cif1);
  hazard1 = cumhaz1 - lag(cumhaz1);
  if _n_ = 1 then do;
    hazard1 = 0;
  end;
run;
* Calculate SD-based CIFs for the other group;
data sd_cif;
  set sd_cif_baseline(keep = time hazard1 rename = (hazard1 = hazard1_baseline));
  hazard1 = hazard1_baseline*&HR1; * HR1 is SD hazard ratio for event of interest;
  if _n_ = 1 then do;
    cumhaz1 = 0;
  end;
  cumhaz1 + hazard1; * Cumulative hazard functions;
  cif1 = 1 - exp(-cumhaz1); * Cumulative incidence functions;
run;
```

Supplemental Figure 1. Examples of SAS programs for implementing cause-specific and sub-distribution hazards models